
2 Correction required for 6.3.4 (IMS enabled on UTRA)
2.1 f_TC_6_3_4_UTRAN 
	Object name 
	f_TC_6_3_4_UTRAN

	Reason for change
	The function f_UTRAN_GMMOnly_Attach(utran_Cell5); is called for the steps 0B to 0M of the expected test sequence. This function call does not take into account the condition when the UE is IMS enabled. This needs to be handled.

After Steps 0B to 0M on UTRAN are done, the control is transferred to the EUTRA PTC when the user manually selects the EUTRA CSG cell.  However, in the current implementation, there is no mechanism to notify the EUTRA PTC that a PDP context has already been established on UTRAN, which can cause issues for the expected signalling messages on EUTRAN. This mechanism needs to be implemented. Please see screenshot

	Summary of change
	Instead of calling f_UTRAN_GMMOnly_Attach(utran_Cell5);, called f_UTRAN_GMMOnly_Attach(utran_Cell5,true);.

Called the function f_IRAT_SendCoOrd to inform the EUTRA PTC about the PDP context status on UTRAN (Whether there is a PDP context activated on UTRAN or not). 

Please see screenshot.  

	Module
	CSG_TestcasesUG_UTRAN.ttcn

	Label
	WA#


Before:
	function f_TC_6_3_4_UTRAN() runs on UTRAN_IRAT_PTC
  {
    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;
    var UtraServingCellPriorityValue_Type v_UtraPriorityValue := 5;
    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := f_UTRAN_ConvertIntegerToEUTRA_MeasurementBandwidth(f_ConvertDL_BandwidthPIXITToInteger());
    //Initialise all cells
    f_UTRAN_Init (EUTRA_UTRAN);
    //Send System Information of UTRAN cell to EUTRA component
    f_UTRAN_Send_IR_SysInfoToEUTRA (utran_Cell5);
    //Receive the system information of EUTRA cell
    v_EUTRASysInfo := f_IRAT_WaitForCoOrd_SysInfo (EUTRA);
    //Update SIB19 of cell 5 according to Table 6.3.4.3.3-4
    f_UTRAN_SysInfo_SetSIB19 (utran_Cell5,
                              cs_SIB19_UtraAndOneEutraEntry (cs_Utra_PriorityInfoList_Def (v_UtraPriorityValue),
                                                             cs_Eutra_FreqAndPriList_OneEntry_Def (v_EUTRASysInfo.Eutra[0].Arfcn,
                                                                                                   v_EUTRA_MeasurementBandwidth)));
    //Create and configure Cell 5
    // @sic R5-155979 sic@
    //Cell started switched on with default power level, -60 and max attenuation
    f_UTRAN_SS_CreateCellFACH (utran_Cell5);
    //Start sending System Information
    f_UTRAN_SendDefSysInfo (utran_Cell5);
    // Wait for CSG list to be emptied in LTE side
    // @sic R5-155979 sic@
    f_IRAT_WaitForCoOrd_Trigger(EUTRA);
    //Set power levels according to T0
    // @sic R5-155979 sic@
    f_UTRAN_SetCellPower(utran_Cell5, -60, -62);
    f_UTRAN_TestBody_Set(true);
    //@siclog "Step 0A" siclog@
    // @sic R5-155979 sic@
    //The UE is switched on.
    f_UT_SwitchOnUE (UT, false);
    //@siclog "Step 0B-0M" siclog@
    // @sic R5-155979 sic@
    //Perform registration procedure
    f_UTRAN_GMMOnly_Attach (utran_Cell5);    
    //Since Authentication parameters have been updated in the registration process, send them to EUTRA PTC
    f_UTRAN_SendAuthCoOrdParameters(EUTRA);        
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_OctetData(f_UTRAN_MobileInfo_EUTRACapability_Get()));    
    f_UTRAN_InterRAT_InitialiseAuthParams();// @sic R5-155979 sic@
<<SKIPPED CODE>>


After:
	function f_TC_6_3_4_UTRAN() runs on UTRAN_IRAT_PTC
  {
    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;
    var UtraServingCellPriorityValue_Type v_UtraPriorityValue := 5;
    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := f_UTRAN_ConvertIntegerToEUTRA_MeasurementBandwidth(f_ConvertDL_BandwidthPIXITToInteger());
    //Initialise all cells
    f_UTRAN_Init (EUTRA_UTRAN);
    //Send System Information of UTRAN cell to EUTRA component
    f_UTRAN_Send_IR_SysInfoToEUTRA (utran_Cell5);
    //Receive the system information of EUTRA cell
    v_EUTRASysInfo := f_IRAT_WaitForCoOrd_SysInfo (EUTRA);
    //Update SIB19 of cell 5 according to Table 6.3.4.3.3-4
    f_UTRAN_SysInfo_SetSIB19 (utran_Cell5,
                              cs_SIB19_UtraAndOneEutraEntry (cs_Utra_PriorityInfoList_Def (v_UtraPriorityValue),
                                                             cs_Eutra_FreqAndPriList_OneEntry_Def (v_EUTRASysInfo.Eutra[0].Arfcn,
                                                                                                   v_EUTRA_MeasurementBandwidth)));
    //Create and configure Cell 5
    // @sic R5-155979 sic@
    //Cell started switched on with default power level, -60 and max attenuation
    f_UTRAN_SS_CreateCellFACH (utran_Cell5);
    //Start sending System Information
    f_UTRAN_SendDefSysInfo (utran_Cell5);
    // Wait for CSG list to be emptied in LTE side
    // @sic R5-155979 sic@
    f_IRAT_WaitForCoOrd_Trigger(EUTRA);
    //Set power levels according to T0
    // @sic R5-155979 sic@
    f_UTRAN_SetCellPower(utran_Cell5, -60, -62);
    f_UTRAN_TestBody_Set(true);
    //@siclog "Step 0A" siclog@
    // @sic R5-155979 sic@
    //The UE is switched on.
    f_UT_SwitchOnUE (UT, false);
    //@siclog "Step 0B-0M" siclog@
    // @sic R5-155979 sic@
    //Perform registration procedure
    f_UTRAN_GMMOnly_Attach (utran_Cell5,true/*WA#6_3_4*/);    
    //Since Authentication parameters have been updated in the registration process, send them to EUTRA PTC
    f_UTRAN_SendAuthCoOrdParameters(EUTRA);
    f_IRAT_SendCoOrd(EUTRA, cms_PDP_Established(vc_UTRAN_Global.PDPEstEarlier));//WA#6_3_4
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_OctetData(f_UTRAN_MobileInfo_EUTRACapability_Get()));    
    f_UTRAN_InterRAT_InitialiseAuthParams();// @sic R5-155979 sic@
<<SKIPPED CODE>>


2.2 f_TC_6_3_4_EUTRA 
	Object name 
	f_TC_6_3_4_EUTRA

	Reason for change
	After preamble on UTRAN is done, the control is transferred to the EUTRA PTC which notifies whether a PDP context is established on UTRAN. However, in the current implementation, there is no mechanism to receive this notification. This needs to be implemented.
In case the testcase is being run with an IMS enabled device, this information needs to be provided to the function f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT so that the correct path in TTCN can be followed for IMS enabled device. 

Also, based upon whether the testcase is run with an IMS enabled device, the call to function f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT needs to be passed different set of parameters. 

This also needs to be done. 

	Summary of change
	Created a variable v_PDPEstEarlier and assigned to this variable the content of the IRAT coordination message.

Also assigned following :

UTRAN.receive (cmr_PDP_Established) -> value v_IRAT_Coordination_MSG;   
v_PDPEstEarlier := v_IRAT_Coordination_MSG.PdpEstEarlier;
Passed the variable v_PDPEstEarlier to the function f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT.
Based upon variable v_PDPEstEarlier called function f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT with different set of arguments. This is done so that the non-IMS path does not get affected.
Please see screenshot.  

	Module
	CSG_TestcasesUG.ttcn

	Label
	WA#


Before:
	function f_TC_6_3_4_EUTRA() runs on EUTRA_PTC
  {
    var template (value) CellPowerList_Type v_CellPowerList_T0;
    var template (value) CellPowerList_Type v_CellPowerList_T0toT1;
    var template (value) CellPowerList_Type v_CellPowerList_T1toT2;
    var template (value) CellPowerList_Type v_CellPowerList_T2toT3;// @sic R5-155979 sic@
    var template (value) CellPowerList_Type v_CellPowerList_T3toT4;// @sic R5-155979 sic@
    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;
    var CellReselectionPriority v_CellReselectionPriority_Cell5 := 5;
    var boolean v_CSInOtherRAT := false;
    var CSG_Identity v_CSG_Identity_Cell2 := '000000000000000000000000001'B;
    var float v_WaitTime := tsc_IdleMode_GenericTimer;
    var IRAT_Coordination_MSG v_IRAT_Coordination_MSG;
    var template (value) SystemInformationBlockType6 v_Sib6;
    var GutiParameters_Type v_GutiParams;// @sic R5-155979 sic@
    timer t_Wait := v_WaitTime;
    v_CellPowerList_T0 := {
      cs_CellPower(eutra_Cell1,tsc_NonSuitableOffCellRS_EPRE),
      cs_CellPower(eutra_Cell2,tsc_NonSuitableOffCellRS_EPRE)
    };
    v_CellPowerList_T0toT1 := {
      cs_CellPower(eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)
    };//@sic R5-155979 sic@
    v_CellPowerList_T1toT2 := {
      cs_CellPower(eutra_Cell1,tsc_NonSuitableOffCellRS_EPRE),
      cs_CellPower(eutra_Cell2,tsc_NonSuitableOffCellRS_EPRE)
    };//@sic R5-155979 sic@
    v_CellPowerList_T2toT3 := {
      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };//@sic R5-155979 sic@
    v_CellPowerList_T3toT4 := {
      cs_CellPower(eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)
    };//@sic R5-155979 sic@
    //Initialise all cells, security and mobile parameters
    f_EUTRA_Init(c11);
    //Set system information combination 4 in cell 1
    v_Sib6 := f_EUTRA_CellInfo_GetSib6(eutra_Cell1); //@sic R5s140094 sic@
    f_EUTRA_ChangeSysinfoCombination (eutra_Cell1, c4);
    f_EUTRA_CellInfo_SetSIB6(eutra_Cell1, v_Sib6); //@sic R5s140094 sic@
    //Coordinate with UTRAN component to receive UTRAN frequencies
    v_UTRANSysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);
    //Send the cell info to the UTRAN component
    f_IRAT_Send_EUTRA_SysInfoToOtherPort(UTRAN, f_EUTRA_CellInfo_GetEARFCN_DL(eutra_Cell1), f_ConvertDL_BandwidthToInteger (f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1)));
    //Set TAC according to 36.523-1 Table 6.0.1-2: Tracking Area Code (TAC) for E-UTRAN cells
    //TAC for Cell 2 is set to 2 in f_EUTRA_Init
    f_EUTRA_CellInfo_SetTAC (eutra_Cell2, tsc_IdleMode_TAC_Cell2);
    //Set Sys. info for Cell 1 and 2 according to Table 6.3.4.3.3-2
    f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell2, true);
    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);
    //Set SIB4 for cell 2 according to Table 6.3.4.3.3-3 and according to default message contents defined in 36.508
    f_EUTRA_CellInfo_SetSIB4 (eutra_Cell2, cs_SIB4_IntraFreqNeighCellList_and_Csg_PhysCellIdRange (cs_IntraFreqNeighCellInfoList_Cell2_ThreeEntriesDef,
                                                                                                   cs_PhysCellIdRange(2, n4)));
    //Set SIB6 for cell 1 and 2 according to Table 6.3.4.3.3-4
    if (f_IRAT_GetUTRAN_FDD_TDD_Mode () == UTRAN_FDD) {
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell1,
                                                  cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq, v_CellReselectionPriority_Cell5),
                                                  omit);
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell2,
                                                  cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq, v_CellReselectionPriority_Cell5),
                                                  omit);
    }
    else {
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell1,
                                                  omit,
                                                  cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq, v_CellReselectionPriority_Cell5));
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell2,
                                                  omit,
                                                  cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq, v_CellReselectionPriority_Cell5));
    }
    // Create and configure Cells
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_Def(eutra_Cell2);
    f_UT_USIM_Insert (UT, "USIM with service n86 available");
    // The UEs Allowed CSG list is empty
    f_EUTRA_EmptyACSGL(eutra_Cell2, {v_CSG_Identity_Cell2}); // @sic R5-155979 R5s160094 sic@
    f_EUTRA_SetCellPowerList (v_CellPowerList_T0);  //@sic R5-155979 sic@
    // Coordination to start preamble in UTRAN
    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);   // @sic R5-155979 sic@
    f_EUTRA_TestBody_Set (true);
    // Since Authentication parameters were updated in the idle updated procedure on UTRAN side...
    //...receive the updated parameters
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);
    UTRAN.receive(cmr_IRAT_OctetData) -> value v_IRAT_Coordination_MSG; // @sic R5-155979 sic@
    // @siclog "Step 0N" siclog@
    // @sic R5-155979 sic@
    // Change Cell 1 and Cell 2 power levels according to the row "T1" in table 6.4.6.3.2-1.
    f_EUTRA_SetCellPowerList (v_CellPowerList_T0toT1);
    // @siclog "Step 0O" siclog@
    // @sic R5-155979 sic@
    // Perform manual CSG selection of cell 2
    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell2);
    f_UT_ManualPLMN_CSG(UT, f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity), v_CSG_Identity_Cell2);
    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT(eutra_Cell2, 
                                                v_CSInOtherRAT, 
                                                false, 
                                                false, 
                                                v_IRAT_Coordination_MSG.OctetData); // @sic R5-155979 sic@
    //@siclog "Step 0P" siclog@
    //@sic R5-155979 sic@
    //Wait 6 seconds for UE to receive system information
    f_Delay (6.0);
    //@siclog "Step 0Q" siclog@
    //@sic R5-155979 sic@
    //Change Cell 1, Cell 2 and Cell 5 levels according to row "T2" in table 6.4.6.3.2-1 and table 6.4.6.3.2-2.
    f_EUTRA_SendAuthParameters(eutra_Cell2, UTRAN);
    f_EUTRA_SetCellPowerList (v_CellPowerList_T1toT2);
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN); //@sic R5s150066 sic@
    //@siclog "Step 1" siclog@
    //The SS changes Cell 1, Cell 2 levels according to the row "T1" in table 6.3.4.3.2-1
    f_EUTRA_SetCellPowerList (v_CellPowerList_T2toT3);  //@sic R5-155979 sic@
    //@siclog "Step 2" siclog@
    //Check: Does the UE send an RRCConnectionRequest on Cell 2?
    f_EUTRA_CheckNoAttachOn1Cell(eutra_Cell2, v_WaitTime, "Test Case 6.3.4 Step 2");
    //@siclog "Step 3" siclog@
    //The SS changes Cell 1, Cell 2 levels according to the row "T2" in table 6.3.4.3.2-1
    f_EUTRA_SetCellPowerList (v_CellPowerList_T3toT4);  //@sic R5-155979 sic@
    t_Wait.start;
    //@siclog "Step 4-10" siclog@
    //Check: Does the test result of steps 1 to 7 of the generic test procedure in TS 36.508 table
    //6.4.2.7A-1 indicate that the UE is camped on E-UTRAN Cell 2 within 6 mins?
    f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT (eutra_Cell2, 
                                                                  v_CSInOtherRAT, 
                                                                  false,

                                                                  true,

                                            v_IRAT_Coordination_MSG.OctetData, 
                                                                  NORMAL);
    t_Wait.stop; // If this times out it will be caught in the default behaviour
    //Tell UTRA PTC that test case is over
    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);
    f_EUTRA_TestBody_Set (false);
    //@sic R5s150804 sic@
    f_EUTRA_SetCellPower(eutra_Cell1,tsc_NonSuitableCellRS_EPRE );
    f_EUTRA_Postamble_EmptyCSG_AllowedList (eutra_Cell2, eutra_Cell2, {v_CSG_Identity_Cell2}, E1_IDLE); //@sic R5s150983 sic@
    //@sic R5s160066 sic@
  } // end of 6.3.4


After:
	function f_TC_6_3_4_EUTRA() runs on EUTRA_PTC
  {
    var template (value) CellPowerList_Type v_CellPowerList_T0;
    var template (value) CellPowerList_Type v_CellPowerList_T0toT1;
    var template (value) CellPowerList_Type v_CellPowerList_T1toT2;
    var template (value) CellPowerList_Type v_CellPowerList_T2toT3;// @sic R5-155979 sic@
    var template (value) CellPowerList_Type v_CellPowerList_T3toT4;// @sic R5-155979 sic@
    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;
    var CellReselectionPriority v_CellReselectionPriority_Cell5 := 5;
    var boolean v_CSInOtherRAT := false;
    var CSG_Identity v_CSG_Identity_Cell2 := '000000000000000000000000001'B;
    var float v_WaitTime := tsc_IdleMode_GenericTimer;
    var IRAT_Coordination_MSG v_IRAT_Coordination_MSG;
    var template (value) SystemInformationBlockType6 v_Sib6;
    var GutiParameters_Type v_GutiParams;// @sic R5-155979 sic@
    var boolean v_PDPEstEarlier := false;//WA#6_3_4
    timer t_Wait := v_WaitTime;
    v_CellPowerList_T0 := {
      cs_CellPower(eutra_Cell1,tsc_NonSuitableOffCellRS_EPRE),
      cs_CellPower(eutra_Cell2,tsc_NonSuitableOffCellRS_EPRE)
    };
    v_CellPowerList_T0toT1 := {
      cs_CellPower(eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)
    };//@sic R5-155979 sic@
    v_CellPowerList_T1toT2 := {
      cs_CellPower(eutra_Cell1,tsc_NonSuitableOffCellRS_EPRE),
      cs_CellPower(eutra_Cell2,tsc_NonSuitableOffCellRS_EPRE)
    };//@sic R5-155979 sic@
    v_CellPowerList_T2toT3 := {
      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE)
    };//@sic R5-155979 sic@
    v_CellPowerList_T3toT4 := {
      cs_CellPower(eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)
    };//@sic R5-155979 sic@
    //Initialise all cells, security and mobile parameters
    f_EUTRA_Init(c11);
    //Set system information combination 4 in cell 1
    v_Sib6 := f_EUTRA_CellInfo_GetSib6(eutra_Cell1); //@sic R5s140094 sic@
    f_EUTRA_ChangeSysinfoCombination (eutra_Cell1, c4);
    f_EUTRA_CellInfo_SetSIB6(eutra_Cell1, v_Sib6); //@sic R5s140094 sic@
    //Coordinate with UTRAN component to receive UTRAN frequencies
    v_UTRANSysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);
    //Send the cell info to the UTRAN component
    f_IRAT_Send_EUTRA_SysInfoToOtherPort(UTRAN, f_EUTRA_CellInfo_GetEARFCN_DL(eutra_Cell1), f_ConvertDL_BandwidthToInteger (f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1)));
    //Set TAC according to 36.523-1 Table 6.0.1-2: Tracking Area Code (TAC) for E-UTRAN cells
    //TAC for Cell 2 is set to 2 in f_EUTRA_Init
    f_EUTRA_CellInfo_SetTAC (eutra_Cell2, tsc_IdleMode_TAC_Cell2);
    //Set Sys. info for Cell 1 and 2 according to Table 6.3.4.3.3-2
    f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell2, true);
    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);
    //Set SIB4 for cell 2 according to Table 6.3.4.3.3-3 and according to default message contents defined in 36.508
    f_EUTRA_CellInfo_SetSIB4 (eutra_Cell2, cs_SIB4_IntraFreqNeighCellList_and_Csg_PhysCellIdRange (cs_IntraFreqNeighCellInfoList_Cell2_ThreeEntriesDef,
                                                                                                   cs_PhysCellIdRange(2, n4)));
    //Set SIB6 for cell 1 and 2 according to Table 6.3.4.3.3-4
    if (f_IRAT_GetUTRAN_FDD_TDD_Mode () == UTRAN_FDD) {
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell1,
                                                  cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq, v_CellReselectionPriority_Cell5),
                                                  omit);
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell2,
                                                  cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq, v_CellReselectionPriority_Cell5),
                                                  omit);
    }
    else {
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell1,
                                                  omit,
                                                  cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq, v_CellReselectionPriority_Cell5));
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell2,
                                                  omit,
                                                  cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq, v_CellReselectionPriority_Cell5));
    }
    // Create and configure Cells
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_Def(eutra_Cell2);
    f_UT_USIM_Insert (UT, "USIM with service n86 available");
    // The UEs Allowed CSG list is empty
    f_EUTRA_EmptyACSGL(eutra_Cell2, {v_CSG_Identity_Cell2}); // @sic R5-155979 R5s160094 sic@
    f_EUTRA_SetCellPowerList (v_CellPowerList_T0);  //@sic R5-155979 sic@
    // Coordination to start preamble in UTRAN
    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);   // @sic R5-155979 sic@
    f_EUTRA_TestBody_Set (true);
    // Since Authentication parameters were updated in the idle updated procedure on UTRAN side...
    //...receive the updated parameters
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);
    UTRAN.receive (cmr_PDP_Established) -> value v_IRAT_Coordination_MSG; //WA#6_3_4     
    v_PDPEstEarlier := v_IRAT_Coordination_MSG.PdpEstEarlier;//WA#6_3_4
    UTRAN.receive(cmr_IRAT_OctetData) -> value v_IRAT_Coordination_MSG; // @sic R5-155979 sic@
    // @siclog "Step 0N" siclog@
    // @sic R5-155979 sic@
    // Change Cell 1 and Cell 2 power levels according to the row "T1" in table 6.4.6.3.2-1.
    f_EUTRA_SetCellPowerList (v_CellPowerList_T0toT1);
    // @siclog "Step 0O" siclog@
    // @sic R5-155979 sic@
    // Perform manual CSG selection of cell 2
    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell2);
    f_UT_ManualPLMN_CSG(UT, f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity), v_CSG_Identity_Cell2);
    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT(eutra_Cell2, 
                                                v_CSInOtherRAT, 
                                                false, 
                                                false, 
                                                v_IRAT_Coordination_MSG.OctetData,
                                                -,//WA#6_3_4
                                                v_PDPEstEarlier/*WA#6_3_4*/); // @sic R5-155979 sic@
    //@siclog "Step 0P" siclog@
    //@sic R5-155979 sic@
    //Wait 6 seconds for UE to receive system information
    f_Delay (6.0);
    //@siclog "Step 0Q" siclog@
    //@sic R5-155979 sic@
    //Change Cell 1, Cell 2 and Cell 5 levels according to row "T2" in table 6.4.6.3.2-1 and table 6.4.6.3.2-2.
    f_EUTRA_SendAuthParameters(eutra_Cell2, UTRAN);
    f_EUTRA_SetCellPowerList (v_CellPowerList_T1toT2);
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN); //@sic R5s150066 sic@
    //@siclog "Step 1" siclog@
    //The SS changes Cell 1, Cell 2 levels according to the row "T1" in table 6.3.4.3.2-1
    f_EUTRA_SetCellPowerList (v_CellPowerList_T2toT3);  //@sic R5-155979 sic@
    //@siclog "Step 2" siclog@
    //Check: Does the UE send an RRCConnectionRequest on Cell 2?
    f_EUTRA_CheckNoAttachOn1Cell(eutra_Cell2, v_WaitTime, "Test Case 6.3.4 Step 2");
    //@siclog "Step 3" siclog@
    //The SS changes Cell 1, Cell 2 levels according to the row "T2" in table 6.3.4.3.2-1
    f_EUTRA_SetCellPowerList (v_CellPowerList_T3toT4);  //@sic R5-155979 sic@
    t_Wait.start;
    //@siclog "Step 4-10" siclog@
    //Check: Does the test result of steps 1 to 7 of the generic test procedure in TS 36.508 table
    //6.4.2.7A-1 indicate that the UE is camped on E-UTRAN Cell 2 within 6 mins?
    //WA#6_3_4 : Reworked this step
    if(v_PDPEstEarlier == true){
            f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT (eutra_Cell2, 
                                                                        v_CSInOtherRAT, 
                                                                        true, //WA#6_3_4 false, 
                                                                        false,//WA#6_3_4 true 
                                                                        v_IRAT_Coordination_MSG.OctetData, 
                                                                        NORMAL,
                                                                        -,//WA#6_3_4
                                                                        v_PDPEstEarlier/*WA#6_3_4*/);
    }else {
            f_EUTRA_508CheckCampOnNewEutraCellUponMobilityFromAnotherRAT (eutra_Cell2, 
                                                                        v_CSInOtherRAT, 
                                                                        true,//WA#6_3_4 false 
                                                                        true, 
                                                                        v_IRAT_Coordination_MSG.OctetData, 
                                                                        NORMAL,
                                                                        -,//WA#6_3_4
                                                                        v_PDPEstEarlier/*WA#6_3_4*/);
    }
    t_Wait.stop; // If this times out it will be caught in the default behaviour
    //Tell UTRA PTC that test case is over
    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);
    f_EUTRA_TestBody_Set (false);
    //@sic R5s150804 sic@
    f_EUTRA_SetCellPower(eutra_Cell1,tsc_NonSuitableCellRS_EPRE );
    f_EUTRA_Postamble_EmptyCSG_AllowedList (eutra_Cell2, eutra_Cell2, {v_CSG_Identity_Cell2}, E1_IDLE); //@sic R5s150983 sic@
    //@sic R5s160066 sic@
  } // end of 6.3.4


2.3 f_EUTRA_TrackingAreaUpdateFromAnotherRAT_WithoutRRCConnReq 
	Object name 
	f_ EUTRA_TrackingAreaUpdateFromAnotherRAT_WithoutRRCConnReq

	Reason for change
	The current implementation of this function doesn’t take into account whether a PDP context was already active on UTRAN. The variable v_EPS_ContextStatus_Rx is set to cr_508_EPS_BearerContextStatusTAUReq which assigns the value ‘001’B to IE ebi5_7. 

The variable v_EPS_ContextStatus_Rx should be updated based on whether a PDP context was active already. This should be able to accommodate both possibilities where the testcase is run with IMS enabled device (in which case there will be a PDP on UTRAN in any case) or with a non-IMS device (in which case there wont be any PDP on UTRAN).
This also needs to be done. 

	Summary of change
	Based on parameter p_PDPContext, assigned the value to IE ebi5_7.

Please note that this change is needed on top of the changes mentioned in CR for testcase 6.2.2.8 (R5sxxyyzz)

Please see screenshot.  

	Module
	EUTRA_CommonProcedures.ttcn

	Label
	WA#


Before:
	function f_EUTRA_TrackingAreaUpdateFromAnotherRAT_WithoutRRCConneReq(EUTRA_CellId_Type p_CellId,
                                                                       RRC_TransactionIdentifier p_RRC_TI,
                                                                       boolean p_CSInOtherRAT,
                                                                       boolean p_PDPContext,
                                                                       boolean p_PreambleOnLTE,
                                                                       template (omit) octetstring p_EUTRA_Cap,   // @sic R5-123734 sic@
                                                                       RRCConnectionReleaseRequired_Type p_ReleaseRequired := rrcConnectionRelease, // @sic R5s150830 sic@
                                                                       EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,
                                                        /*WA#6_2_2_8*/ boolean p_PDPEstEarlierBecauseOfIMSRegUtra := false) runs on EUTRA_PTC return template (omit) NasCount_Type
  { /* @sic R5s120027 Additional Changes sic@ */
    var NAS_MSG_Indication_Type v_NasInd;
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var template (omit) NasCount_Type v_NasCountUL := omit; // @sic R5s120944 sic@
    var NasCountInfo_Type v_NasCountInfo;
    var SRB_COMMON_IND v_ReceivedAsp;
    var GutiParameters_Type v_GutiParams:= f_EUTRA_CellInfo_GetGuti(p_CellId);
    var B3_Type v_EpsUpdate_Result := f_GetEPSTAUType(p_ForcedAttach); // @sic R5s100029, R5s110176 sic@
    var template (present) B3_Type v_EpsUpdate_TypeValue := v_EpsUpdate_Result; // To be used in Request message
    var template (present) EPS_BearerContextStatus v_EPS_ContextStatus_Rx := cr_508_EPS_BearerContextStatusTAUReq; // @sic R5s100608, R5-110746 sic@
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus_Tx;
    var template (present) NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5s110176 sic@
    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType (p_ForcedAttach); // @sic R5s110176 sic@
    var template (present) NAS_MSG_Indication_Type v_NAS_Expected, v_TAU_Expected, v_TAU_NoProtection, v_TAU_StartInOtherRAT, v_TAU_ForcedMapped;
    var boolean v_NoAuthSincePowerOn;
    var template (omit) AccessStratumRelease v_Rel := omit;
    var UE_EUTRA_Capability v_EUTRA_Cap;
    var boolean v_Rel8_9 := false;
    var  template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (p_ForcedAttach);
    var B3_Type v_PDNType := f_GetPdnType();
    var template CiphKeySeqNum v_CiphKeySeqNum := cr_CiphKeySeqNum('8'H, ?) ifpresent;
    timer t_WaitforAttach := f_EUTRA_SetTimerToleranceMax( p_CellId, nonProtocolTimer, tsc_WaitforAttach);  // @sic R5-134358 sic@
    if (p_PreambleOnLTE) { // @sic R5-123734 sic@
      v_Rel := f_EUTRA_MobileInfo_GetUECapability_AccessStratumRelease ();
    } else {
      if (ispresent(p_EUTRA_Cap)) {
        v_EUTRA_Cap := f_EUTRA_DecodeEutraCapMsg(valueof(p_EUTRA_Cap)); // This needs to be a template as it can be omitted
        v_Rel := v_EUTRA_Cap.accessStratumRelease;
      }
    }
    if (not p_PDPContext and not p_PreambleOnLTE) { // no bearer has previously been established
      v_EPS_ContextStatus_Rx.ebi5_7 := '000'B; // so set all bearer indications to 0
    }

    if (not match(tsc_EpsUpdate_TaUpdate, v_EpsUpdate_TypeValue)) { // TA Only;  @sic R5-110746 sic@
      if (p_CSInOtherRAT) {
        v_EpsUpdate_TypeValue := tsc_EpsUpdate_Combined_IMSIAttach;
      } else {
        v_EpsUpdate_TypeValue := (tsc_EpsUpdate_Combined_TaLaUpdate, tsc_EpsUpdate_Combined_IMSIAttach); // @sic R5s120943, R5-140325 sic@
      }
    }
<<SKIPPED CODE>>


After:
	function f_EUTRA_TrackingAreaUpdateFromAnotherRAT_WithoutRRCConneReq(EUTRA_CellId_Type p_CellId,
                                                                       RRC_TransactionIdentifier p_RRC_TI,
                                                                       boolean p_CSInOtherRAT,
                                                                       boolean p_PDPContext,
                                                                       boolean p_PreambleOnLTE,
                                                                       template (omit) octetstring p_EUTRA_Cap,   // @sic R5-123734 sic@
                                                                       RRCConnectionReleaseRequired_Type p_ReleaseRequired := rrcConnectionRelease, // @sic R5s150830 sic@
                                                                       EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,
                                                        /*WA#6_2_2_8*/ boolean p_PDPEstEarlierBecauseOfIMSRegUtra := false) runs on EUTRA_PTC return template (omit) NasCount_Type
  { /* @sic R5s120027 Additional Changes sic@ */
    var NAS_MSG_Indication_Type v_NasInd;
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var template (omit) NasCount_Type v_NasCountUL := omit; // @sic R5s120944 sic@
    var NasCountInfo_Type v_NasCountInfo;
    var SRB_COMMON_IND v_ReceivedAsp;
    var GutiParameters_Type v_GutiParams:= f_EUTRA_CellInfo_GetGuti(p_CellId);
    var B3_Type v_EpsUpdate_Result := f_GetEPSTAUType(p_ForcedAttach); // @sic R5s100029, R5s110176 sic@
    var template (present) B3_Type v_EpsUpdate_TypeValue := v_EpsUpdate_Result; // To be used in Request message
    var template (present) EPS_BearerContextStatus v_EPS_ContextStatus_Rx := cr_508_EPS_BearerContextStatusTAUReq; // @sic R5s100608, R5-110746 sic@
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus_Tx;
    var template (present) NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5s110176 sic@
    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType (p_ForcedAttach); // @sic R5s110176 sic@
    var template (present) NAS_MSG_Indication_Type v_NAS_Expected, v_TAU_Expected, v_TAU_NoProtection, v_TAU_StartInOtherRAT, v_TAU_ForcedMapped;
    var boolean v_NoAuthSincePowerOn;
    var template (omit) AccessStratumRelease v_Rel := omit;
    var UE_EUTRA_Capability v_EUTRA_Cap;
    var boolean v_Rel8_9 := false;
    var  template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (p_ForcedAttach);
    var B3_Type v_PDNType := f_GetPdnType();
    var template CiphKeySeqNum v_CiphKeySeqNum := cr_CiphKeySeqNum('8'H, ?) ifpresent;
    timer t_WaitforAttach := f_EUTRA_SetTimerToleranceMax( p_CellId, nonProtocolTimer, tsc_WaitforAttach);  // @sic R5-134358 sic@
    if (p_PreambleOnLTE) { // @sic R5-123734 sic@
      v_Rel := f_EUTRA_MobileInfo_GetUECapability_AccessStratumRelease ();
    } else {
      if (ispresent(p_EUTRA_Cap)) {
        v_EUTRA_Cap := f_EUTRA_DecodeEutraCapMsg(valueof(p_EUTRA_Cap)); // This needs to be a template as it can be omitted
        v_Rel := v_EUTRA_Cap.accessStratumRelease;
      }
    }
    if (not p_PDPContext and not p_PreambleOnLTE) { // no bearer has previously been established
      v_EPS_ContextStatus_Rx.ebi5_7 := '000'B; // so set all bearer indications to 0
    }
    //WA#6_3_4 : Variable p_PDPContext is being used to manage flow for both URRAN and EUTRAN
    // If a PDP context is already active on UTRAN, the IE ebi5_7 would be set to '011'B
    //However, when there is no PDP on UTRAN and UE is not IMS enabled, IE ebi5_7 would be set to '001'B
    // To handle both cases, following is implemented.
    if(p_PDPContext)
    {
    
v_EPS_ContextStatus_Rx.ebi5_7 := ('001'B,'011'B);
    }    
    if (not match(tsc_EpsUpdate_TaUpdate, v_EpsUpdate_TypeValue)) { // TA Only;  @sic R5-110746 sic@
      if (p_CSInOtherRAT) {
        v_EpsUpdate_TypeValue := tsc_EpsUpdate_Combined_IMSIAttach;
      } else {
        v_EpsUpdate_TypeValue := (tsc_EpsUpdate_Combined_TaLaUpdate, tsc_EpsUpdate_Combined_IMSIAttach); // @sic R5s120943, R5-140325 sic@
      }
    }
<<SKIPPED CODE>>


